This worksheet demonstrates the study of structural properties of linear systems defined by a time-varying ordinary differential equations.

See E. D. Sontag, *Mathematical Control Theory: Deterministic Finite Dimensional Systems*, Springer, 1990, p. 114.

```
> with(Ore_algebra):
> with(OreModules):
> Alg := DefineOreAlgebra(diff=[D,t], polynom=[t]):
```

Let us consider the following matrix of ordinary differential operators:

```
> R := evalm([[D, -t]]);
```

$$R := \begin{bmatrix} \mathrm{D} & -t \end{bmatrix}$$

The matrix R corresponds to the following time-varying ordinary differential linear system:

```
> ApplyMatrix(R, [x(t),u(t)], Alg)[1,1]=0;
```

$$(\tfrac{d}{dt}\,\mathrm{x}(t)) - t\,\mathrm{u}(t) = 0$$

Let us check whether or not this system is controllable and flat. In order to do that, let us define the formal adjoint $R\_\ adj$ of R.

```
> R_adj := Involution(R, Alg);
```

$$R\_adj := \begin{bmatrix} -\mathrm{D} \\ -t \end{bmatrix}$$

We compute the first extension module with values in Alg of the left Alg-module which is associated with R_adj:

```
> Ext := Exti(R_adj, Alg, 1);
```

$$Ext := [\begin{bmatrix} 1 \end{bmatrix}, \begin{bmatrix} \mathrm{D} & -t \end{bmatrix}, \begin{bmatrix} -t^2 & -1 + t\,\mathrm{D} \\ -2 - t\,\mathrm{D} & \mathrm{D}^2 \end{bmatrix}]$$

Therefore, we obtain that the left Alg-module M associated with R is torsion-free, and thus, projective as Alg is a hereditary ring. A parametrization of the system is given by $Ext[3]$. This result can directly be obtained by using the following command:

```
> Parametrization(R, Alg);
```

$$\begin{bmatrix} -t^2\,\xi_1(t) - \xi_2(t) + t\,(\frac{d}{dt}\,\xi_2(t)) \\ -2\,\xi_1(t) - t\,(\frac{d}{dt}\,\xi_1(t)) + (\frac{d^2}{dt^2}\,\xi_2(t)) \end{bmatrix}$$

Let us notice that the previous parametrization depends on two arbitrary functions $\xi 1$ and $\xi 2$. However, the system has only 1 input, and thus, the rank of the left Alg-module M is 1. Let us check this result:

```
> OreRank(R, Alg);
```

$$1$$

Hence, we deduce that there exist some minimal parametrizations of the system which depend on 1 arbitrary function. Let us compute some of them.

```
> P := MinimalParametrizations(R, Alg);
```

$$P := \left[\begin{bmatrix} -t^2 \\ -2 - t\,\mathrm{D} \end{bmatrix}, \begin{bmatrix} -1 + t\,\mathrm{D} \\ \mathrm{D}^2 \end{bmatrix}\right]$$

Let us check whether or not the first minimal parametrization $P[1]$ is injective.

```
> LeftInverse(P[1], Alg);
```

$$[]$$

We obtain that $P[1]$ is not an injective parametrization of the system. Let us examine the second minimal parametrization $P[2]$ in a similar way:

```
> LeftInverse(P[2], Alg);
```

$$[]$$

We find that $P[2]$ is not an injective parametrization of the system. Therefore, we cannot conclude that the left *Alg*-module M associated with the system is free. In fact, we can prove that M is a stably free but not a free left *Alg*-module. See A. Quadrat, D. Robertz, "*On Monge problem for uncontrollable linear systems*", to appear. However, we already know that M is a projective left *Alg*-module. This result can also be obtained by checking whether or not a right-inverse of R exists.

```
> RightInverse(R, Alg);
```

$$\begin{bmatrix} t \\ \mathrm{D} \end{bmatrix}$$

Finally, one of the main interests of the non-minimal parametrization *Ext*[3] is that it admits a generalized inverse, namely there exists a matrix G with entries in *Alg* such that *Ext*[3] G *Ext*[3] = *Ext*[3] (contrary to $P[1]$ and $P[2]$). Let us compute one generalized inverse of *Ext*[3]:

```
> G := GeneralizedInverse(Ext[3], Alg);
```

$$G := \begin{bmatrix} 0 & -1 \\ -1 & 0 \end{bmatrix}$$

```
> Mult(Ext[3], G, Ext[3], Alg);
```

$$\begin{bmatrix} -t^2 & -1 + t\,\mathrm{D} \\ -2 - t\,\mathrm{D} & \mathrm{D}^2 \end{bmatrix}$$

Let us determine the obstruction of flatness. In order to do that, we study the system over the ring *AlgRat* of ordinary differential operators with rational coefficients in t. Let us compute a parametrization of the system by allowing to invert non-zero polynomials in t. We obtain:

```
> Extrat := ExtiRat(R_adj, Alg, 1);
```

$$Extrat := \left[\begin{bmatrix} 1 \end{bmatrix}, \begin{bmatrix} \mathrm{D} & -t \end{bmatrix}, \begin{bmatrix} -t^2 \\ -2 - t\,\mathrm{D} \end{bmatrix}\right]$$

In particular, we obtain that the left *AlgRat*-module M is torsion-free, and thus, free because *AlgRat* is a left principal ideal domain.

Moreover, a (minimal) parametrization of the system is defined by *Extrat*[3]. This result can directly be obtained by using *ParametrizationRat*:

```
> ParametrizationRat(R, Alg);
```

$$\begin{bmatrix} -t^2\,\xi_1(t) \\ -2\,\xi_1(t) - t\,(\frac{d}{dt}\,\xi_1(t)) \end{bmatrix}$$

The fact that the left *AlgRat*-module M associated with R is free implies that, away from some singularities that we are going to determine, the system is flat. Let us compute a basis for this module.

```
> S := LeftInverseRat(Extrat[3], Alg);
```

$$S := \begin{bmatrix} -\frac{1}{t^2} & 0 \end{bmatrix}$$

We obtain that a basis of the left *AlgRat*-module M is defined by $\xi = (S\;x\;:\;u)^T$ and satisfies

$$(x : u)^T = Extrat[3]\,\xi.$$

In particular, we see that this parametrization is not defined for $t = 0$ as we have a singularity. Therefore, the system is flat except for $t = 0$. Finally, let us notice that, away from $t = 0$, we have another right-inverse of R defined by:

```
> RightInverseRat(R, Alg);
```

$$\begin{bmatrix} 0 \\ -\frac{1}{t} \end{bmatrix}$$

Let us compute the Brunovský canonical form:

```
> B := BrunovskyRat(R, Alg);
```

$$B := \begin{bmatrix} -\frac{1}{t^2} & 0 \\ \frac{2}{t^3} & -\frac{1}{t} \end{bmatrix}$$

Let us check that the variables z and v defined by $(z\;:\;v)^T = B\;(x\;:\;u)^T$ satisfy a Brunovský canonical form:

```
> E := EliminationRat(linalg[stackmatrix](B, R), [x,u], [z,v,0], Alg):
> ApplyMatrix(E[1], [x(t),u(t)], Alg)=ApplyMatrix(E[2], [z(t),v(t)], Alg);
```

$$\begin{bmatrix} 0 \\ \mathrm{u}(t) \\ \mathrm{x}(t) \end{bmatrix} = \begin{bmatrix} -(\frac{d}{dt}\,\mathrm{z}(t)) + \mathrm{v}(t) \\ -2\,\mathrm{z}(t) - t\,\mathrm{v}(t) \\ -t^2\,\mathrm{z}(t) \end{bmatrix}$$

The first equation shows that z and v satisfy a Brunovský canonical form. The last two equations give x and u in terms of z and v.

We continue with another example of a linear system defined by ordinary differential equations.

See E. D. Sontag, *Mathematical Control Theory: Deterministic Finite Dimensional Systems*, Springer, 1990, p. 114.

Let us consider the following matrix of ordinary differential operators:

```
> R2 := evalm([[D-t, -1, 0, 0],[0, D-t^3, 0, -1], [0, 0, D-t^2, -1]]);
```

$$R2 := \begin{bmatrix} \mathrm{D} - t & -1 & 0 & 0 \\ 0 & \mathrm{D} - t^3 & 0 & -1 \\ 0 & 0 & \mathrm{D} - t^2 & -1 \end{bmatrix}$$

The time-varying ordinary differential system associated with $R2$ is defined by:

```
> ApplyMatrix(R2, [x1(t),x2(t),x3(t),u(t)], Alg) = evalm([[0],[0],[0]]);
```

$$\begin{bmatrix} -t\,\mathrm{x1}(t) + (\frac{d}{dt}\,\mathrm{x1}(t)) - \mathrm{x2}(t) \\ -t^3\,\mathrm{x2}(t) + (\frac{d}{dt}\,\mathrm{x2}(t)) - \mathrm{u}(t) \\ -t^2\,\mathrm{x3}(t) + (\frac{d}{dt}\,\mathrm{x3}(t)) - \mathrm{u}(t) \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix}$$

Let us study whether or not this system is controllable and flat.

```
> R2_adj := Involution(R2, Alg);
```

$$R2\_adj := \begin{bmatrix} -\mathrm{D} - t & 0 & 0 \\ -1 & -\mathrm{D} - t^3 & 0 \\ 0 & 0 & -\mathrm{D} - t^2 \\ 0 & -1 & -1 \end{bmatrix}$$

```
> ext := Exti(R2_adj, Alg, 1):  ext[1];
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

We obtain that over the Weyl algebra $Alg = A_1$, the left Alg-module $M2$ which is associated with $R2$ is torsion-free, and thus, projective. In particular, the system is controllable and a parametrization is given by $ext[3]$.

```
> map(collect, ext[3], D);
```

$[-1597\,\mathrm{D}^3 + (-163 + 4050\,t + 1597\,t^2)\,\mathrm{D}^2$
$+ (13380 - 6683\,t - 2339\,t^2 - 4050\,t^3)\,\mathrm{D} + 7726\,t^2 + 24955 - 13423\,t + 2502\,t^4$
$- 1302\,t^3, (-7816\,t + 2951)\,\mathrm{D}^2 + (-24485 + 10984\,t^2 - 3575\,t + 3025\,t^3)\,\mathrm{D}$
$- 12892 + 87086\,t - 60739\,t^2 + 4791\,t^5 + 3575\,t^3 - 13935\,t^4, 4791\,\mathrm{D}^3$
$+ (-3845 + 19\,t)\,\mathrm{D}^2 + (-2800 - 11938\,t^2 + 57950\,t + 4772\,t^3 - 4791\,t^4)\,\mathrm{D}$
$+ 60406\,t^2 - 17306 - 100922\,t + 15783\,t^4 - 4791\,t^5 - 458\,t^3]$

$[-1597\,\mathrm{D}^4 + (1597\,t^2 + 5647\,t - 163)\,\mathrm{D}^3$
$+ (17430 - 6389\,t^2 - 3326\,t - 5647\,t^3)\,\mathrm{D}^2$
$+ (-31481\,t + 2259\,t^2 + 6552\,t^4 + 1037\,t^3 + 18272)\,\mathrm{D} + 2282\,t^3 - 2502\,t^5 - 13423$
$- 9503\,t + 9517\,t^2 + 1302\,t^4, (-7816\,t + 2951)\,\mathrm{D}^3$
$+ (18800\,t^2 - 6526\,t - 32301 + 3025\,t^3)\,\mathrm{D}^2$
$+ (-48089\,t^2 + 133539\,t - 7409\,t^3 - 16467 - 16960\,t^4 + 4791\,t^5)\,\mathrm{D} + 13935\,t^5$
$+ 87086 - 108586\,t + 4999\,t^3 - 76361\,t^2 + 20380\,t^4 - 4791\,t^6, 4791\,\mathrm{D}^4$
$+ (-4772\,t - 3845)\,\mathrm{D}^3 + (-11957\,t^2 - 4791\,t^4 + 61795\,t - 2781 + 4772\,t^3)\,\mathrm{D}^2$
$+ (-121998\,t + 16772\,t^2 + 11011\,t^4 - 7684\,t^3 + 40644)\,\mathrm{D} + 2726\,t^3 - 15783\,t^5$
$- 100922 + 138118\,t + 99548\,t^2 - 23497\,t^4 + 4791\,t^6]$

$[-1597\,\mathrm{D}^4 + (-163 + 5647\,t + 1597\,t^3)\,\mathrm{D}^3$
$+ (-132\,t + 163\,t^3 - 11343\,t^2 + 17430 - 5647\,t^4)\,\mathrm{D}^2$
$+ (2259\,t^2 - 21573\,t + 6063\,t^3 + 6552\,t^5 + 15078 - 1465\,t^4)\,\mathrm{D} + 1302\,t^5 - 23331$
$- 9503\,t + 9517\,t^2 - 5495\,t^3 + 2282\,t^4 - 2502\,t^6, (-7816\,t + 2951)\,\mathrm{D}^3$
$+ (-6526\,t + 21751\,t^2 - 7742\,t^3 - 32301 + 7816\,t^4)\,\mathrm{D}^2$
$+ (127637\,t - 48089\,t^2 - 21751\,t^5 - 12898\,t^3 + 4791\,t^6 + 11317\,t^4 - 16467)\,\mathrm{D}$
$+ 92988 - 108586\,t - 76361\,t^2 - 4791\,t^7 + 20380\,t^5 - 28538\,t^4 + 35995\,t^3$
$+ 13935\,t^6, 4791\,\mathrm{D}^4 + (-3845 - 4791\,t^3 - 4772\,t + 4791\,t^2)\,\mathrm{D}^3$
$+ (4772\,t^4 + 52213\,t - 1429\,t^2 - 2781 - 4791\,t^5 + 3845\,t^3)\,\mathrm{D}^2$
$+ (50226 + 9058\,t^3 - 143054\,t + 16772\,t^2 + 15802\,t^5 - 28258\,t^4)\,\mathrm{D} - 79866$
$- 23497\,t^5 + 138118\,t + 36263\,t^4 + 99548\,t^2 + 4791\,t^7 - 15783\,t^6 - 23896\,t^3]$

$[-1597\,\mathrm{D}^5 + (5647\,t - 163 + 1597\,t^2 + 1597\,t^3)\,\mathrm{D}^4$
$+ (-132\,t + 23077 - 5484\,t^3 - 5647\,t^4 - 6389\,t^2 - 1597\,t^5)\,\mathrm{D}^3$
$+ (-14682\,t^2 + 14946 - 16393\,t^3 + 6389\,t^5 - 44259\,t + 5647\,t^6 + 9878\,t^4)\,\mathrm{D}^2+$
$(10218\,t^3 - 4985\,t + 32783\,t^4 - 4761\,t^5 + 12628\,t^2 - 1037\,t^6 - 44904 - 6552\,t^7)\,\mathrm{D}$
$- 9503 + 2502\,t^8 + 19034\,t + 18631\,t^3 + 6846\,t^2 - 3007\,t^4 - 1302\,t^7 - 2282\,t^6$
$- 9517\,t^5, (-7816\,t + 2951)\,\mathrm{D}^4$
$+ (7816\,t^4 + 74\,t^3 + 18800\,t^2 - 40117 - 6526\,t)\,\mathrm{D}^3$
$+ (171139\,t - 22993 - 14009\,t^5 - 3025\,t^6 - 39014\,t^2 - 10434\,t^4 + 24892\,t^3)\,\mathrm{D}^2+$
$(-4791\,t^8 - 89204\,t^4 + 220625 - 98588\,t^2 - 204764\,t + 62024\,t^5 - 46374\,t^3$
$+ 2618\,t^6 + 16960\,t^7)\mathrm{D} - 108586 - 13935\,t^8 - 152722\,t - 5566\,t^3 + 14997\,t^2$
$+ 4791\,t^9 - 20380\,t^7 - 4999\,t^6 + 178261\,t^4 + 47615\,t^5, 4791\,\mathrm{D}^5$
$+ (-4772\,t - 3845 - 4791\,t^3)\,\mathrm{D}^4$
$+ (-11957\,t^2 - 19\,t^4 + 61795\,t - 7553 + 8617\,t^3)\,\mathrm{D}^3 + (31088\,t^2 + 4791\,t^7$
$- 145912\,t + 102439 - 4772\,t^6 + 11957\,t^5 - 50784\,t^4 - 24067\,t^3)\mathrm{D}^2 + (76496\,t^2$
$+ 6126\,t^3 - 222920 + 98501\,t^4 + 171662\,t + 12475\,t^6 - 32555\,t^5 - 11011\,t^7)\mathrm{D}$
$+ 138118 - 4791\,t^9 + 199096\,t + 23497\,t^7 + 8178\,t^2 - 70802\,t^5 + 15783\,t^8$
$- 2726\,t^6 - 217033\,t^4 + 6934\,t^3]$

However, *ext*[3] is not an injective parametrization as we have:

```
>  LeftInverse(ext[3], Alg);
```

$$[]$$

The fact that *M2* is a projective left *Alg*-module implies that there exists a right-inverse of *R2*. Let us compute one:

```
>  T := map(collect, RightInverse(R2, Alg), D);
```

$T :=$

$$\Big[(-\frac{190\,t}{4791} + \frac{122}{4791})\,\mathrm{D}^2 + (-\frac{746}{4791} + \frac{211}{4791}\,t - \frac{395}{4791}\,t^2 + \frac{190}{4791}\,t^3)\,\mathrm{D} + \frac{584}{4791} - \frac{1681\,t}{4791}$$
$$- \frac{394\,t^2}{4791} - \frac{211\,t^3}{4791} + \frac{91\,t^4}{1597}, (\frac{541\,t}{9582} + \frac{608}{4791})\,\mathrm{D}^2$$
$$+ (\frac{2225}{9582} - \frac{2845}{9582}\,t - \frac{5}{4791}\,t^2 - \frac{541}{9582}\,t^3)\,\mathrm{D} - \frac{5572}{4791} + \frac{7802\,t}{4791} + \frac{1021\,t^2}{9582} + \frac{2845\,t^3}{9582}$$
$$- \frac{201\,t^4}{1597}, (-\frac{541\,t}{9582} - \frac{608}{4791})\,\mathrm{D}^2 + (-\frac{2225}{9582} + \frac{2845}{9582}\,t + \frac{5}{4791}\,t^2 + \frac{541}{9582}\,t^3)\,\mathrm{D} + \frac{5572}{4791}$$
$$- \frac{7802\,t}{4791} - \frac{1021\,t^2}{9582} - \frac{2845\,t^3}{9582} + \frac{201\,t^4}{1597}\Big]$$

$$\Big[(-\frac{190\,t}{4791} + \frac{122}{4791})\,\mathrm{D}^3 + (-\frac{312}{1597} - \frac{205}{4791}\,t^2 + \frac{89}{4791}\,t + \frac{190}{4791}\,t^3)\,\mathrm{D}^2$$
$$+ (\frac{265}{1597} - \frac{575}{1597}\,t + \frac{184}{4791}\,t^3 - \frac{35}{4791}\,t^2 + \frac{83}{4791}\,t^4)\,\mathrm{D} - \frac{6472}{4791} - \frac{91\,t^5}{1597} - \frac{1372\,t}{4791}$$
$$+ \frac{1486\,t^3}{4791} + \frac{1048\,t^2}{4791} + \frac{211\,t^4}{4791}, (\frac{541\,t}{9582} + \frac{608}{4791})\,\mathrm{D}^3$$
$$+ (\frac{461}{1597} - \frac{551}{9582}\,t^2 - \frac{4061}{9582}\,t - \frac{541}{9582}\,t^3)\,\mathrm{D}^2$$
$$+ (-\frac{4663}{3194} + \frac{4453}{3194}\,t + \frac{2855}{9582}\,t^3 + \frac{2243}{9582}\,t^2 - \frac{665}{9582}\,t^4)\,\mathrm{D} + \frac{7802}{4791} + \frac{201\,t^5}{1597} + \frac{6593\,t}{4791}$$
$$- \frac{5845\,t^3}{9582} - \frac{7069\,t^2}{9582} - \frac{2845\,t^4}{9582}, (-\frac{541\,t}{9582} - \frac{608}{4791})\,\mathrm{D}^3$$
$$+ (-\frac{461}{1597} + \frac{551}{9582}\,t^2 + \frac{4061}{9582}\,t + \frac{541}{9582}\,t^3)\,\mathrm{D}^2$$
$$+ (\frac{4663}{3194} - \frac{4453}{3194}\,t - \frac{2855}{9582}\,t^3 - \frac{2243}{9582}\,t^2 + \frac{665}{9582}\,t^4)\,\mathrm{D} - \frac{7802}{4791} - \frac{201\,t^5}{1597} - \frac{6593\,t}{4791}$$
$$+ \frac{5845\,t^3}{9582} + \frac{7069\,t^2}{9582} + \frac{2845\,t^4}{9582}\Big]$$

$$\Big[(-\frac{190\,t}{4791} + \frac{122}{4791})\,\mathrm{D}^3 + (\frac{89}{4791}\,t - \frac{83}{4791}\,t^2 - \frac{312}{1597} - \frac{122}{4791}\,t^3 + \frac{190}{4791}\,t^4)\,\mathrm{D}^2$$
$$+ (-\frac{1969}{4791}\,t - \frac{35}{4791}\,t^2 + \frac{449}{4791}\,t^3 + \frac{83}{4791}\,t^5 + \frac{265}{1597} - \frac{89}{4791}\,t^4)\,\mathrm{D} - \frac{2076}{1597} - \frac{1372\,t}{4791}$$
$$+ \frac{1048\,t^2}{4791} - \frac{590\,t^3}{4791} + \frac{211\,t^5}{4791} + \frac{1486\,t^4}{4791} - \frac{91\,t^6}{1597}, (\frac{541\,t}{9582} + \frac{608}{4791})\,\mathrm{D}^3$$
$$+ (-\frac{4061}{9582}\,t + \frac{665}{9582}\,t^2 + \frac{461}{1597} - \frac{608}{4791}\,t^3 - \frac{541}{9582}\,t^4)\,\mathrm{D}^2$$
$$+ (\frac{10927}{9582}\,t + \frac{3517}{4791}\,t^2 - \frac{904}{4791}\,t^3 - \frac{665}{9582}\,t^5 - \frac{4663}{3194} + \frac{4061}{9582}\,t^4)\,\mathrm{D} + \frac{3006}{1597} + \frac{1802\,t}{4791}$$
$$- \frac{7069\,t^2}{9582} + \frac{167\,t^3}{9582} - \frac{2845\,t^5}{9582} - \frac{5845\,t^4}{9582} + \frac{201\,t^6}{1597}, (-\frac{541\,t}{9582} - \frac{608}{4791})\,\mathrm{D}^3$$
$$+ (\frac{4061}{9582}\,t - \frac{665}{9582}\,t^2 - \frac{461}{1597} + \frac{608}{4791}\,t^3 + \frac{541}{9582}\,t^4)\,\mathrm{D}^2$$
$$+ (-\frac{10927}{9582}\,t - \frac{3517}{4791}\,t^2 + \frac{904}{4791}\,t^3 + \frac{665}{9582}\,t^5 + \frac{4663}{3194} - \frac{4061}{9582}\,t^4)\,\mathrm{D} - \frac{3006}{1597} - \frac{1802\,t}{4791}$$
$$+ \frac{7069\,t^2}{9582} - \frac{167\,t^3}{9582} + \frac{2845\,t^5}{9582} + \frac{5845\,t^4}{9582} - \frac{201\,t^6}{1597}\Big]$$

$$\Big[(-\frac{190\,t}{4791}+\frac{122}{4791})\,\mathrm{D}^4+(\frac{190}{4791}\,t^4+\frac{68}{4791}\,t^3-\frac{205}{4791}\,t^2-\frac{1126}{4791}+\frac{89}{4791}\,t)\,\mathrm{D}^3$$
$$+(-\frac{2135}{4791}\,t-\frac{190}{4791}\,t^6+\frac{205}{4791}\,t^5+\frac{884}{4791}+\frac{535}{4791}\,t^2-\frac{2}{1597}\,t^4+\frac{1120}{4791}\,t^3)\,\mathrm{D}^2$$
$$+(-\frac{238}{4791}\,t^5+\frac{1936}{4791}\,t^4+\frac{341}{1597}\,t^3+\frac{1600}{4791}\,t^2-\frac{1442}{4791}\,t-\frac{8197}{4791}-\frac{184}{4791}\,t^6-\frac{83}{4791}\,t^7)\,\mathrm{D}$$
$$+\frac{7316\,t^3}{4791}+\frac{2096\,t}{4791}-\frac{1048\,t^5}{4791}+\frac{1486\,t^2}{1597}-\frac{211\,t^7}{4791}-\frac{1486\,t^6}{4791}+\frac{7\,t^4}{4791}+\frac{91\,t^8}{1597}$$
$$-\frac{1372}{4791},(\frac{541\,t}{9582}+\frac{608}{4791})\,\mathrm{D}^4+(-\frac{541}{9582}\,t^4-\frac{1757}{9582}\,t^3-\frac{551}{9582}\,t^2+\frac{3307}{9582}-\frac{4061}{9582}\,t)\,\mathrm{D}^3$$
$$+(\frac{12257}{9582}\,t+\frac{541}{9582}\,t^6+\frac{551}{9582}\,t^5-\frac{9025}{4791}+\frac{310}{4791}\,t^2+\frac{566}{1597}\,t^4+\frac{89}{9582}\,t^3)\,\mathrm{D}^2$$
$$+(-\frac{1037}{9582}\,t^5-\frac{8102}{4791}\,t^4+\frac{914}{1597}\,t^3+\frac{748}{4791}\,t^2+\frac{8836}{4791}\,t+\frac{28963}{9582}-\frac{2855}{9582}\,t^6+\frac{665}{9582}\,t^7)\,\mathrm{D}$$
$$+\frac{1802}{4791}-\frac{7069\,t}{4791}+\frac{7069\,t^5}{9582}-\frac{5845\,t^2}{3194}+\frac{2845\,t^7}{9582}+\frac{5845\,t^6}{9582}-\frac{3578\,t^4}{4791}-\frac{201\,t^8}{1597}$$
$$-\frac{13492\,t^3}{4791},(-\frac{541\,t}{9582}-\frac{608}{4791})\,\mathrm{D}^4$$
$$+(\frac{541}{9582}\,t^4+\frac{1757}{9582}\,t^3+\frac{551}{9582}\,t^2-\frac{3307}{9582}+\frac{4061}{9582}\,t)\,\mathrm{D}^3$$
$$+(-\frac{12257}{9582}\,t-\frac{541}{9582}\,t^6-\frac{551}{9582}\,t^5+\frac{9025}{4791}-\frac{310}{4791}\,t^2-\frac{566}{1597}\,t^4-\frac{89}{9582}\,t^3)\,\mathrm{D}^2$$
$$+(\frac{1037}{9582}\,t^5+\frac{8102}{4791}\,t^4-\frac{914}{1597}\,t^3-\frac{748}{4791}\,t^2-\frac{8836}{4791}\,t-\frac{28963}{9582}+\frac{2855}{9582}\,t^6-\frac{665}{9582}\,t^7)\,\mathrm{D}$$
$$-\frac{6593}{4791}+\frac{7069\,t}{4791}-\frac{7069\,t^5}{9582}+\frac{5845\,t^2}{3194}-\frac{2845\,t^7}{9582}-\frac{5845\,t^6}{9582}+\frac{3578\,t^4}{4791}+\frac{201\,t^8}{1597}$$
$$+\frac{13492\,t^3}{4791}\Big]$$

In fact, we have:

```
>  Mult(R2, T, Alg);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

The fact that *R2* admits a right-inverse implies that *ext*[3] admits a generalized inverse. Let us compute it.

```
>  g := GeneralizedInverse(ext[3], Alg);
```

$g :=$

$$\left[\frac{245131}{4628992335}, -\frac{541\,\mathrm{D}}{15302454} - \frac{2231213107177}{23611647424230030}, \frac{4835593673}{73925007589950}, 0\right]$$

$$\left[\frac{4288}{114768405} - \frac{38\,\mathrm{D}}{2898555} - \frac{38\,t}{2898555},\right.$$
$$-\frac{665}{45907362}t^2 - \frac{78163}{13886977005}\mathrm{D} - \frac{444284147}{44355004553970}t - \frac{59395081751}{585412746055290},$$
$$\left.\frac{95918282}{916425713925} - \frac{541\,\mathrm{D}}{14492775} - \frac{3172237\,t}{138869770050}, 0\right]$$

$$\left[\frac{25769}{1262452455} - \frac{38\,t}{2898555},\right.$$
$$\frac{541}{45907362}t^2 - \frac{541}{45907362}\mathrm{D} + \frac{198835379}{22177502276985}t - \frac{419732645023}{6439540206608190},$$
$$\left.\frac{1437199157}{20161365706350} - \frac{6820387\,t}{138869770050}, 0\right]$$

We easily check that we have *ext*[3] *g* *ext*[3] - *ext*[3] = 0:

```
> simplify(Mult(ext[3], g, ext[3], Alg) - ext[3]);
```

$$0$$

```
> linalg[coldim](ext[3]);
```

$$3$$

The parametrization *ext*[3] depends on 3 arbitrary functions. However, the rank of the left *Alg*-module *M2* is:

```
> OreRank(R2, Alg);
```

$$1$$

Indeed, it agrees with the number of inputs, i.e., 1. Therefore, we can obtain some minimal parametrizations which involve only 1 arbitrary function:

```
> Q := MinimalParametrizations(R2, Alg):
> nops(Q);
```

$$3$$

We obtain 3 minimal parametrizations of the system. The first one is defined by:

```
> map(collect, Q[1], D);
```

$$[-1597\,\mathrm{D}^3 + (-163 + 4050\,t + 1597\,t^2)\,\mathrm{D}^2$$
$$+ (13380 - 6683\,t - 2339\,t^2 - 4050\,t^3)\,\mathrm{D} + 7726\,t^2 + 24955 - 13423\,t + 2502\,t^4$$
$$- 1302\,t^3]$$
$$[-1597\,\mathrm{D}^4 + (1597\,t^2 + 5647\,t - 163)\,\mathrm{D}^3$$
$$+ (17430 - 6389\,t^2 - 3326\,t - 5647\,t^3)\,\mathrm{D}^2$$
$$+ (-31481\,t + 2259\,t^2 + 6552\,t^4 + 1037\,t^3 + 18272)\,\mathrm{D} + 2282\,t^3 - 2502\,t^5 - 13423$$
$$- 9503\,t + 9517\,t^2 + 1302\,t^4]$$
$$[-1597\,\mathrm{D}^4 + (-163 + 5647\,t + 1597\,t^3)\,\mathrm{D}^3$$
$$+ (-132\,t + 163\,t^3 - 11343\,t^2 + 17430 - 5647\,t^4)\,\mathrm{D}^2$$
$$+ (2259\,t^2 - 21573\,t + 6063\,t^3 + 6552\,t^5 + 15078 - 1465\,t^4)\,\mathrm{D} + 1302\,t^5 - 23331$$
$$- 9503\,t + 9517\,t^2 - 5495\,t^3 + 2282\,t^4 - 2502\,t^6]$$
$$[-1597\,\mathrm{D}^5 + (5647\,t - 163 + 1597\,t^2 + 1597\,t^3)\,\mathrm{D}^4$$
$$+ (-132\,t + 23077 - 5484\,t^3 - 5647\,t^4 - 6389\,t^2 - 1597\,t^5)\,\mathrm{D}^3$$
$$+ (-14682\,t^2 + 14946 - 16393\,t^3 + 6389\,t^5 - 44259\,t + 5647\,t^6 + 9878\,t^4)\,\mathrm{D}^2 +$$
$$(10218\,t^3 - 4985\,t + 32783\,t^4 - 4761\,t^5 + 12628\,t^2 - 1037\,t^6 - 44904 - 6552\,t^7)\,\mathrm{D}$$
$$- 9503 + 2502\,t^8 + 19034\,t + 18631\,t^3 + 6846\,t^2 - 3007\,t^4 - 1302\,t^7 - 2282\,t^6$$
$$- 9517\,t^5]$$

The second one is defined by:

```
>  map(collect, Q[2], D);
```

$$[(-7816\,t + 2951)\,\mathrm{D}^2 + (-24485 + 10984\,t^2 - 3575\,t + 3025\,t^3)\,\mathrm{D} - 12892$$
$$+ 87086\,t - 60739\,t^2 + 4791\,t^5 + 3575\,t^3 - 13935\,t^4]$$
$$[(-7816\,t + 2951)\,\mathrm{D}^3 + (18800\,t^2 - 6526\,t - 32301 + 3025\,t^3)\,\mathrm{D}^2$$
$$+ (-48089\,t^2 + 133539\,t - 7409\,t^3 - 16467 - 16960\,t^4 + 4791\,t^5)\,\mathrm{D} + 13935\,t^5$$
$$+ 87086 - 108586\,t + 4999\,t^3 - 76361\,t^2 + 20380\,t^4 - 4791\,t^6]$$
$$[(-7816\,t + 2951)\,\mathrm{D}^3 + (-6526\,t + 21751\,t^2 - 7742\,t^3 - 32301 + 7816\,t^4)\,\mathrm{D}^2$$
$$+ (127637\,t - 48089\,t^2 - 21751\,t^5 - 12898\,t^3 + 4791\,t^6 + 11317\,t^4 - 16467)\,\mathrm{D}$$
$$+ 92988 - 108586\,t - 76361\,t^2 - 4791\,t^7 + 20380\,t^5 - 28538\,t^4 + 35995\,t^3$$
$$+ 13935\,t^6]$$
$$[(-7816\,t + 2951)\,\mathrm{D}^4 + (7816\,t^4 + 74\,t^3 + 18800\,t^2 - 40117 - 6526\,t)\,\mathrm{D}^3$$
$$+ (171139\,t - 22993 - 14009\,t^5 - 3025\,t^6 - 39014\,t^2 - 10434\,t^4 + 24892\,t^3)\,\mathrm{D}^2 +$$
$$(-4791\,t^8 - 89204\,t^4 + 220625 - 98588\,t^2 - 204764\,t + 62024\,t^5 - 46374\,t^3$$
$$+ 2618\,t^6 + 16960\,t^7)\mathrm{D} - 108586 - 13935\,t^8 - 152722\,t - 5566\,t^3 + 14997\,t^2$$
$$+ 4791\,t^9 - 20380\,t^7 - 4999\,t^6 + 178261\,t^4 + 47615\,t^5]$$

The last one is defined by:

```
>  map(collect, Q[3], D);
```

$$[4791\,\mathrm{D}^3 + (-3845 + 19\,t)\,\mathrm{D}^2$$
$$+ (-2800 - 11938\,t^2 + 57950\,t + 4772\,t^3 - 4791\,t^4)\,\mathrm{D} + 60406\,t^2 - 17306$$
$$- 100922\,t + 15783\,t^4 - 4791\,t^5 - 458\,t^3]$$
$$[4791\,\mathrm{D}^4 + (-4772\,t - 3845)\,\mathrm{D}^3$$
$$+ (-11957\,t^2 - 4791\,t^4 + 61795\,t - 2781 + 4772\,t^3)\,\mathrm{D}^2$$
$$+ (-121998\,t + 16772\,t^2 + 11011\,t^4 - 7684\,t^3 + 40644)\,\mathrm{D} + 2726\,t^3 - 15783\,t^5$$
$$- 100922 + 138118\,t + 99548\,t^2 - 23497\,t^4 + 4791\,t^6]$$
$$[4791\,\mathrm{D}^4 + (-3845 - 4791\,t^3 - 4772\,t + 4791\,t^2)\,\mathrm{D}^3$$
$$+ (4772\,t^4 + 52213\,t - 1429\,t^2 - 2781 - 4791\,t^5 + 3845\,t^3)\,\mathrm{D}^2$$
$$+ (50226 + 9058\,t^3 - 143054\,t + 16772\,t^2 + 15802\,t^5 - 28258\,t^4)\,\mathrm{D} - 79866$$
$$- 23497\,t^5 + 138118\,t + 36263\,t^4 + 99548\,t^2 + 4791\,t^7 - 15783\,t^6 - 23896\,t^3]$$
$$[4791\,\mathrm{D}^5 + (-4772\,t - 3845 - 4791\,t^3)\,\mathrm{D}^4$$
$$+ (-11957\,t^2 - 19\,t^4 + 61795\,t - 7553 + 8617\,t^3)\,\mathrm{D}^3 + (31088\,t^2 + 4791\,t^7$$
$$- 145912\,t + 102439 - 4772\,t^6 + 11957\,t^5 - 50784\,t^4 - 24067\,t^3)\mathrm{D}^2 + (76496\,t^2$$
$$+ 6126\,t^3 - 222920 + 98501\,t^4 + 171662\,t + 12475\,t^6 - 32555\,t^5 - 11011\,t^7)\mathrm{D}$$
$$+ 138118 - 4791\,t^9 + 199096\,t + 23497\,t^7 + 8178\,t^2 - 70802\,t^5 + 15783\,t^8$$
$$- 2726\,t^6 - 217033\,t^4 + 6934\,t^3]$$

We can check that none of the previous minimal parametrizations $Q[i]$ admits a left-inverse:

```
> seq(LeftInverse(Q[i], Alg), i=1..3);
```

$$[], [], []$$

Now, let us consider the *AlgRat*-module *M2* associated with *R2* over the ring *AlgRat* of ordinary differential operators with rational coefficients.

```
> extrat := ExtiRat(R2_adj, Alg, 1):
> extrat[1]; map(collect, extrat[3], D);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

$$[\%1\,\mathrm{D} - 4 - 2\,t^6 + t^5 - 6\,t^2 + 12\,t + t^7 + 18\,t^3 - 13\,t^4]$$
$$[\%1\,\mathrm{D}^2 + (-6 + 23\,t^3 + t^7 - 17\,t^4 - t^6 - t^5 - 7\,t^2 + 18\,t)\,\mathrm{D} - 13\,t^4 + 6\,t^6 - 46\,t^3 + 12$$
$$+ t^5 - t^8 + 2\,t^7 - 8\,t + 42\,t^2]$$
$$[\%1\,\mathrm{D}^2 + (-6 + 18\,t - 5\,t^5 + 21\,t^3 - 2\,t^7 + t^8 + 2\,t^6 - 12\,t^4 - 7\,t^2)\,\mathrm{D} + 12 - 8\,t + 2\,t^8$$
$$+ 42\,t^2 - 18\,t^4 - t^9 - 8\,t^6 - 42\,t^3 + 11\,t^5 + 6\,t^7]$$
$$[\%1\,\mathrm{D}^3 + (t^8 + 24\,t - 4\,t^5 + 31\,t^3 - t^7 - 8 - 10\,t^2 - 20\,t^4)\,\mathrm{D}^2$$
$$+ (30 + 19\,t^7 - 22\,t + t^9 - 36\,t^4 - t^{10} + 2\,t^5 - 108\,t^3 - 10\,t^6 + 111\,t^2)\,\mathrm{D} - 8 + 60\,t^6$$
$$+ 84\,t - 6\,t^9 - 138\,t^2 + t^{11} + 13\,t^4 - t^8 - 64\,t^3 - 6\,t^5 - 2\,t^{10} + 5\,t^7]$$
$$\%1 := -t^5 - t^3 + 2\,t^4 - 2\,t + 3\,t^2$$

We obtain that the *Algrat*-module *M2* associated with *R2* is torsion-free, and thus, free as *AlgRat* is a left principal ideal domain.

A parametrization of the system is then given by *extrat*[3]. Let us check whether or not it is injective.

```
> S := LeftInverseRat(extrat[3], Alg);
```

$$S := \left[\begin{array}{cccc} \frac{t-1}{\%1} & -\frac{1}{t^2\,\%1} & \frac{1}{t^2\,\%1} & 0 \end{array}\right]$$

$$\%1 := 4 + 13\,t^2 - 4\,t^7 + 6\,t^6 - 12\,t - 14\,t^3 + t^8 - 10\,t^5 + 17\,t^4$$

```
> Mult(S, extrat[3], Alg);
```

$$\left[\begin{array}{c} 1 \end{array}\right]$$

Therefore, we obtain that *extrat*[3] is an injective parametrization, and thus, a basis of the *AlgRat*-module *M2* is defined by $\xi = S\ (x1\ :\ x2\ :\ x3\ :\ u)^T$, where ξ satisfies $(x1\ :\ x2\ :\ x3\ :\ u)^T = extrat[3]\ \xi$. Let us compute the singularities of S.

```
> map(factor, S);
```

$$\left[\begin{array}{cccc} \frac{t-1}{(t-2)^2\,(t^3+t-1)^2} & -\frac{1}{t^2\,(t-2)^2\,(t^3+t-1)^2} & \frac{1}{t^2\,(t-2)^2\,(t^3+t-1)^2} & 0 \end{array}\right]$$

In particular, we see that $t = 0$ is a singularity of S, and thus, *extrat*[3] does not admit a left-inverse at $t = 0$.

More generally, we finally obtain that the system defined by *R2* is flat expect for the possible real values $t = 0$ and 2. To finish, let us compute the Brunovský canonical form of the system:

```
> B2 := BrunovskyRat(R2, Alg);
```

$B2 :=$

$$\left[\frac{t-1}{\%4}, -\frac{1}{t^2\,\%4}, \frac{1}{t^2\,\%4}, 0\right]$$

$$\left[\frac{5\,t + 14\,t^3 - 4 - 10\,t^2 - 3\,t^5 - 4\,t^4 + t^6}{\%3},\right.$$
$$-\frac{-4 - 2\,t^6 + t^5 - 6\,t^2 + 12\,t + t^7 + 18\,t^3 - 13\,t^4}{t^3\,\%3},$$
$$\left.\frac{-4 - 2\,t^6 + t^5 - 6\,t^2 + 12\,t + t^7 + 18\,t^3 - 13\,t^4}{t^3\,\%3}, 0\right]$$

$$\left[\frac{-184\,t^4 + 97\,t^3 - 21\,t^2 + 6\,t - 26 - 23\,t^7 - 127\,t^6 + 232\,t^5 + t^{11} + 43\,t^8 - 5\,t^{10} - 3\,t^9}{\%2}\right.$$
$$, -(-456\,t^5 - 264\,t^3 + 502\,t^4 - 417\,t^7 + 445\,t^6 + 120\,t^8 - 120\,t + 24 + 29\,t^{11} + 2\,t^9$$
$$- 23\,t^{12} - 3\,t^{10} + 228\,t^2 + t^{15} + 11\,t^{13} - 5\,t^{14})/(t^4\,\%2), (-98\,t^9 + 24 - 522\,t^5$$
$$+ 228\,t^2 + 538\,t^4 - 272\,t^3 - 534\,t^7 + 532\,t^6 + 243\,t^8 - 120\,t - 28\,t^{11} + 81\,t^{10} + 6\,t^{12}$$
$$\left.- 4\,t^{13} + t^{14})/(t^4\,\%2), 0\right]$$

$$\left[(-593\,t^3 - 120\,t^2 + 126\,t - 300 - 3300\,t^6 - 162\,t^5 + 1640\,t^4 + 988\,t^{10} + 5256\,t^7\right.$$
$$+ 365\,t^9 - 3591\,t^8 + 73\,t^{13} - 91\,t^{12} - 432\,t^{11} + 3\,t^{14} - 7\,t^{15} + t^{16})/(\%2\,\%1), -($$
$$15892\,t^5 + 6720\,t^3 - 10272\,t^4 + 18602\,t^7 - 16558\,t^6 - 19052\,t^8 + 1296\,t - 192$$
$$+ 5215\,t^{11} + 15305\,t^9 + 283\,t^{12} - 11903\,t^{10} - 3720\,t^2 - 871\,t^{15} - 1112\,t^{13}$$
$$+ 1114\,t^{14} + 294\,t^{16} - 7\,t^{17} - 54\,t^{20} - 103\,t^{18} + 105\,t^{19} + t^{23} + 22\,t^{21} - 7\,t^{22})/(t^5\,\%2$$
$$\%1), (24100\,t^9 - 192 + 17364\,t^5 - 3720\,t^2 - 10752\,t^4 + 6784\,t^3 + 23562\,t^7$$
$$- 19446\,t^6 - 26448\,t^8 + 1296\,t + 14448\,t^{11} - 21300\,t^{10} - 7147\,t^{12} + 894\,t^{15}$$
$$+ 4233\,t^{13} - 2427\,t^{14} + 189\,t^{17} - 384\,t^{16} + 15\,t^{19} - 53\,t^{18} - 6\,t^{20} + t^{21})/(t^5\,\%2\,\%1),$$
$$\left.-\frac{1}{t\,\%1}\right]$$

$$\%1 := t^4 + t^2 - 2\,t^3 + 2 - 3\,t$$
$$\%2 := 16 - 96\,t + 150\,t^{12} + 248\,t^2 + 442\,t^{10} - 68\,t^{13} - 8\,t^{15} + 28\,t^{14} + t^{16} - 912\,t^7 + 926\,t^6$$
$$+ 641\,t^4 - 636\,t^9 + 829\,t^8 - 424\,t^3 - 852\,t^5 - 284\,t^{11}$$
$$\%3 := 8 - 36\,t + t^{12} + 66\,t^2 + 15\,t^{10} - 84\,t^7 + 100\,t^6 + 117\,t^4 - 29\,t^9 + 57\,t^8 - 87\,t^3 - 123\,t^5$$
$$- 6\,t^{11}$$
$$\%4 := 4 + 13\,t^2 - 4\,t^7 + 6\,t^6 - 12\,t - 14\,t^3 + t^8 - 10\,t^5 + 17\,t^4$$

Let us check that the variables $z1$, $z2$, $z3$ and v defined by $(z1 : z2 : z3 : v)^T = B2\ (x1 : x2 : x3 : u)^T$ satisfy a Brunovský canonical form.

```
> E2 := EliminationRat(linalg[stackmatrix](B2, R2), [x1,x2,x3,u],
> [z1,z2,z3,v,0,0,0], Alg):
> ApplyMatrix(E2[1], [x1(t),x2(t),x3(t),u(t)], Alg)=
> ApplyMatrix(E2[2], [z1(t),z2(t),z3(t),v(t)], Alg);
```

$$\begin{bmatrix} 0 \\ 0 \\ 0 \\ \mathrm{u}(t) \\ \mathrm{x3}(t) \\ \mathrm{x2}(t) \\ \mathrm{x1}(t) \end{bmatrix} =$$

$$\left[-(\tfrac{d}{dt}\,\mathrm{z3}(t)) + \mathrm{v}(t)\right]$$
$$\left[-(\tfrac{d}{dt}\,\mathrm{z2}(t)) + \mathrm{z3}(t)\right]$$
$$\left[-(\tfrac{d}{dt}\,\mathrm{z1}(t)) + \mathrm{z2}(t)\right]$$
$$[(t^{11} - 2\,t^{10} - 6\,t^9 - t^8 + 5\,t^7 + 60\,t^6 - 6\,t^5 + 13\,t^4 - 64\,t^3 - 138\,t^2 + 84\,t - 8)\,\mathrm{z1}(t)$$
$$+ (-t^{10} + t^9 + 19\,t^7 - 10\,t^6 + 2\,t^5 - 36\,t^4 - 108\,t^3 + 111\,t^2 - 22\,t + 30)\,\mathrm{z2}(t)$$
$$+ (t^8 - t^7 - 4\,t^5 - 20\,t^4 + 31\,t^3 - 10\,t^2 + 24\,t - 8)\,\mathrm{z3}(t) + \%1\,\mathrm{v}(t)]$$
$$[(-t^9 + 2\,t^8 + 6\,t^7 - 8\,t^6 + 11\,t^5 - 18\,t^4 - 42\,t^3 + 42\,t^2 - 8\,t + 12)\,\mathrm{z1}(t)$$
$$+ (-6 + 18\,t - 5\,t^5 + 21\,t^3 - 2\,t^7 + t^8 + 2\,t^6 - 12\,t^4 - 7\,t^2)\,\mathrm{z2}(t) + \%1\,\mathrm{z3}(t)]$$
$$[(-t^8 + 2\,t^7 + 6\,t^6 + t^5 - 13\,t^4 - 46\,t^3 + 42\,t^2 - 8\,t + 12)\,\mathrm{z1}(t)$$
$$+ (-7\,t^2 - t^5 + 18\,t + t^7 + 23\,t^3 - 17\,t^4 - t^6 - 6)\,\mathrm{z2}(t) + \%1\,\mathrm{z3}(t)]$$
$$[(-4 - 2\,t^6 + t^5 - 6\,t^2 + 12\,t + t^7 + 18\,t^3 - 13\,t^4)\,\mathrm{z1}(t) + \%1\,\mathrm{z2}(t)]$$
$$\%1 := -t^5 + 2\,t^4 - t^3 + 3\,t^2 - 2\,t$$

The first three equations show that *z1*, *z2*, *z3* and v satisfy a Brunovský canonical form. Moreover, the last four equations give *x1*, *x2*, *x3* and u in terms of *z1*, *z2*, *z3* and v.

Finally, for time-varying ordinary differential systems, we demonstrate that we generally need to compute more terms in the time-varying Kalman's criterion than in the time-invariant Kalman's criterion for controllability. We compute column by column the controllability matrix which is needed for Kalman's criterion.

For the system studied above, we have:

```
> F := evalm([[t, 1, 0],[0, t^3, 0], [0, 0, t^2]]);
```

$$F := \begin{bmatrix} t & 1 & 0 \\ 0 & t^3 & 0 \\ 0 & 0 & t^2 \end{bmatrix}$$

```
> G := evalm([[0],[1],[1]]);
```

$$G := \begin{bmatrix} 0 \\ 1 \\ 1 \end{bmatrix}$$

The first three columns of the controllability matrix are given by:

```
> C3 := ControllabilityMatrix(F, G, 3, Alg);
```

$$C3 := \begin{bmatrix} 0 & 1 & t + t^3 \\ 1 & t^3 & t^6 - 3\,t^2 \\ 1 & t^2 & t^4 - 2\,t \end{bmatrix}$$

We check again the singularities of the system:

```
> factor(linalg[det](C3));
```

$$t\,(t-2)\,(t^3 + t - 1)$$

```
> linalg[rank](subs(t=0, evalm(C3)));
```

$$2$$

```
> linalg[rank](subs(t=2, evalm(C3)));
```

$$2$$

```
> linalg[rank](subs(t=1, evalm(C3)));
```

$$3$$

Hence, at the singularities Kalman's controllability matrix does not have maximal rank. Let us continue:

```
> C4 := ControllabilityMatrix(F, G, 4, Alg);
```

$$C4 := \begin{bmatrix} 0 & 1 & t+t^3 & -5\,t^2+t^4+t^6-1 \\ 1 & t^3 & t^6-3\,t^2 & t^9-9\,t^5+6\,t \\ 1 & t^2 & t^4-2\,t & t^6-6\,t^3+2 \end{bmatrix}$$

```
> linalg[rank](subs(t=0, evalm(C4)));
```

$$3$$

```
> linalg[rank](subs(t=2, evalm(C4)));
```

$$3$$

Now the controllability matrix *C4* has maximal rank, and thus, we find again that the system defined by *R2* is controllable. Note that it was not sufficient to form the matrix $(G,\, F\,G + ...,\, F^2\,G + ...)$ as defined in the classical Kalman test ($n = 3$), but we had to investigate the matrix $(G,\, F\,G + ...,\, F^2\,G + ...,\, F^3\,G + ...)$. For more details, see E. D. Sontag, *Mathematical Control Theory: Deterministic Finite Dimensional Systems*, Springer, 1990.